

## How Prolog Handles a Query

Trace it by hand

### Example 1

Database:

```
1) male(tom).
2) male(peter).
3) male(doug).
4) female(susan).
5) male(david).

6) parent(doug, susan).
7) parent(tom, william).
8) parent(doug, david).
9) parent(doug, tom).

10) grandfather(GP, GC) :- male(GP),
    parent(GP, X),
    parent(X, GC).
```

Query:

```
?- grandfather(X,Y).
```

22

23

### Trace it in Prolog

```
[trace] ?- grandfather(X,Y).
  Call: (7) grandfather(_G283, _G284) ? creep
  Call: (8) male(_G283) ? creep
  Exit: (8) male(tom) ? creep
  Call: (8) parent(tom, _L205) ? creep
  Exit: (8) parent(tom, william) ? creep
  Call: (8) parent(william, _G284) ? creep
  Fail: (8) parent(william, _G284) ? creep
  Redo: (8) male(_G283) ? creep
  Exit: (8) male(peter) ? creep
  Call: (8) parent(peter, _L205) ? creep
  Fail: (8) parent(peter, _L205) ? creep
  Redo: (8) male(_G283) ? creep
  Exit: (8) male(doug) ? creep
  Call: (8) parent(doug, _L205) ? creep
  Exit: (8) parent(doug, susan) ? creep
  Call: (8) parent(susan, _G284) ? creep
  Fail: (8) parent(susan, _G284) ? creep
  Redo: (8) parent(doug, _L205) ? creep
  Exit: (8) parent(doug, david) ? creep
  Call: (8) parent(david, _G284) ? creep
  Fail: (8) parent(david, _G284) ? creep
  Redo: (8) parent(doug, _L205) ? creep
  Exit: (8) parent(doug, tom) ? creep
  Call: (8) parent(tom, _G284) ? creep
  Exit: (8) parent(tom, william) ? creep
  Exit: (7) grandfather(doug, william) ? creep
X = doug
Y = william
Yes
```

24

### Prolog Search Trees

- Each **node** is an ordered list of goals.
- Each **edge** is labelled with the variable bindings that occurred due to applying a rule. (The binding are in effect throughout the subtree.)
- Each **leaf** represents either success or failure.

25

## Example 2

## Trace it by hand

Database:

```
1) male(albert).
2) female(alice).
3) male(edward).
4) female(victoria).
5) parent(albert,edward).
6) parent(victoria,edward).
7) parent(albert,alice).
8) parent(victoria,alice).
9) sibling(X, Y) :- parent(P, X), parent(P, Y).
```

Query:

```
?- sibling(alice,Asib).
Asib = edward ;
Asib = alice ;
Asib = edward;
Asib = alice ;
No
?- sibling(Asib, alice).
Asib = edward ;
Asib = edward ;
Asib = alice ;
Asib = alice ;
No
```

26

27

## Trace it in Prolog

```
[trace] ?- sibling(alice,Asib).
  Call: (7) sibling(alice, _G284) ? creep
  Call: (8) parent(_L205, alice) ? creep
  Exit: (8) parent(albert, alice) ? creep
  Call: (8) parent(albert, _G284) ? creep
  Exit: (8) parent(albert, edward) ? creep
  Exit: (7) sibling(alice, edward) ? creep

Asib = edward ;
  Redo: (8) parent(albert, _G284) ? creep
  Exit: (8) parent(albert, alice) ? creep
  Exit: (7) sibling(alice, alice) ? creep

Asib = alice ;
  Redo: (8) parent(_L205, alice) ? creep
  Exit: (8) parent(victoria, alice) ? creep
  Call: (8) parent(victoria, _G284) ? creep
  Exit: (8) parent(victoria, edward) ? creep
  Exit: (7) sibling(alice, edward) ? creep

Asib = edward ;
  Redo: (8) parent(victoria, _G284) ? creep
  Exit: (8) parent(victoria, alice) ? creep
  Exit: (7) sibling(alice, alice) ? creep

Asib = alice ;

No
```

28

## The Anonymous Variable

If a rule has a variable that appears only once, that variable is called a “singleton variable”.

Its value doesn't matter — it doesn't have to match anything elsewhere in the rule.

```
isaMother(X) :- female(X), parent(X, Y).
```

Such a variable consumes resources at run time.

We can replace it with “\_”, the anonymous variable. It matches anything.

If we don't, Prolog will warn us.

29

## Procedural Semantics of Prolog

Notice the recursion in this algorithm: “find” calls “find”. This reasoning is recursively applied until we reach rules that are facts.

This process is called **Backward Chaining**.

30

## Logic Programming vs. Prolog

```
cousin(X,Y) :- parent(W,X), sister(W,Z),
              parent(Z,Y).
```

```
cousin(X,Y) :- parent(W,X), brother(W,Z),
              parent(Z,Y).
```

```
?- cousin(X,jane). % a query
```

Rule and Goal Ordering:

- There are two rules for cousin
- Which rule do we try first?
- Each rule for cousin has several subgoals
- Which subgoal do we try first?

31

## Logic Programming vs. Prolog

**Logic Programming:** *Nondeterministic*

- Arbitrarily choose rule to expand first
- Arbitrarily choose subgoal to explore first
- Results don't depend on rule and subgoal ordering

**Prolog:** *Deterministic*

- Expand first rule first
- Explore first subgoal first
- Results may depend on rule and subgoal ordering

32

## Let's Write Some Code

Write these predicates:

1. uncle
2. aunt
3. nephew
4. niece
5. grandparent
6. ancestor

Code we develop in class will be posted on the Web site as text so that you can load and run it.

33

## Transitive Relations

```
parent(sally,jane).    parent(bob,jane).
parent(sally,john).   parent(bob,john).
parent(mary,sally).   parent(al,sally).
parent(ann,bob).      parent(mike,bob).
parent(jean,al).      parent(joe,al).
parent(ruth,mary).    parent(jim,mary).
parent(esther,ruth).  parent(mick,ruth).

grandparent(X,Y) :- parent(X,Z), parent(Z,Y).
ancestor(X,Y) :- parent(X,Y).
ancestor(X,Y) :- parent(X,Z), ancestor(Z,Y).
```

```
?- grandparent(Y,jane).
```

```
Y = mary ;
```

```
Y = al ;
```

```
Y = ann ;
```

```
Y = mike ;
```

```
No
```

```
?- ancestor(X,jane).
```

```
X = sally ;
```

```
X = bob ;
```

```
X = mary ;
```

```
X = al ;
```

```
X = ann ;
```

```
X = mike ;
```

```
X = jean ;
```

```
X = joe ;
```

```
X = ruth ;
```

```
X = jim ;
```

```
X = esther ;
```

```
X = mick ;
```

```
No
```